

**\*\*\* SUBROUTINAS PARA LEER Y ESCRIBIR DATOS EN LA EEPROM DEL 16F88 \*\*\***

;subrutina para leer dato de EEPROM 16F88  
;para usar esta subrutina, debe escribirse antes la dirección deseada  
;en el registro EEADR (BANCO2). La rutina regresa con el dato en w.

```
EERD:   BANCO3
        BCF      EECON1,EEPGD ;APUNTA HACIA EEPROM DE DATOS
        BSF      EECON1,RD    ;HABILITA EL BIT 0 (RD) DE EECON1
        BANCO2
        MOVF     EEDATA,W      ;TRANSFIERE EL DATO EN EEDATA A W,
        RETURN
```

;subrutina para escribir dato en EEPROM 16F88  
;para usar esta subrutina, debe escribirse antes la dirección DE LA EEPROM  
;en el registro EEADR (BANCO2) y el dato en el registro EEDATA (BANCO2)  
;la rutina regresa una vez que el dato fue escrito en la localidad deseada.

```
EEWR:   BANCO3 ;HABILITA EL BANCO 3
        BCF      EECON1,EEPGD ;APUNTA HACIA EEPROM DE DATOS
        BSF      EECON1,WREN  ;HABILITA ESCRITURA EN EEPROM
        BCF      INTCON, GIE  ;DESHABILITA INTERRUPCIONES
        MOVLW    H'55'        ;PREPARA SECUENCIA DE SEGURIDAD
        MOVWF    EECON2       ;ESCRIBE PRIMER DATO DE SECUENCIA
        MOVLW    H'AA'        ;SEGUNDO DATO
        MOVWF    EECON2       ;ESCRIBE SEGUNDO DATO DE SECUENCIA
        BSF      EECON1,WR    ;INICIA CICLO DE ESCRITURA
EW:     BTFSC    EECON1,WR    ;MALLA PARA ESPERAR AL FINAL DEL CICLO
        GOTO     EW           ;SI WR=1, CICLO DE ESCRITURA AUN NO TERMINA
        BCF      EECON1, WREN ;DESHABILITA ESCRITURA
        BSF      INTCON, GIE  ;HABILITA INTERRUPCIONES
        RETURN
```